

## Sprawozdanie końcowe projekt WSYZ

Michał Sadowski 325221

Mateusz Ostaszewski 325203

Krzysztof Gólc 325159

### 1. Zadanie:

Rozważana jest produkcja i dystrybucja podstawowych warzyw, tj. ziemniaków, kapusty, buraków i marchwi w Warszawie i okolicach. Istnieją trzy rodzaje przedsiębiorstw:

- Grupa 6 producentów: P1...P6. Każdy z producentów produkuje każdy rodzaj warzyw jednak w różnych maksymalnych ilościach rocznych podanych w poniższej tabeli [tony]:

Producenci:

	Ziemniaki	Kapusta	Buraki	Marchew
Błonie	100	180	120	160
Książenice	60	90	150	50
Góra Kalwaria	160	70	190	90
Otwock	260	50	60	100
Wołomin	250	230	190	110
Legionowo	200	140	60	90

- Sieć 3 magazynów-chłodni: M1..M3. Każdy magazyn ma określoną pojemność wyrażoną w tonach (800, 1200, 750) i może służyć do przechowywania dowolnych warzyw.

Magazyny:

	Pojemność
Pruszków	800
Piaseczno	1200
Zielonka	750

- Sieć sklepów spożywczych usytuowanych w Warszawie (proszę zaproponować 10 sklepów rozlokowanych w różnych punktach Warszawy (adres i pozycja GPS)).


Sklepy:

Nazwa	Adres	Pojemność
Biedronka	Grzybowska 32, 00-863 Warszawa	8.5
Auchan Warszawa Okęcie	Aleja Krakowska 61, 02-183 Warszawa	8.78
Carrefour	Plac Unii, Puławska 2, 02- 566, Warszawa	7.44
Lidl	Aleja Rzeczypospolitej 29E, 02-972 Warszawaa	8.36
Delikatesy Santorini	Kwatery Głównej 46, 04-294 Warszawa	8.14
He-man	Księdza Ignacego Skorupki 17, 05-091 Ząbki	8.38
Delikatesy Centrum	Skarbka z Gór 57, 03-287 Warszawa	8.3
Agora	Przy Agorze 15, 01-960 Warszawa	8.68
Arhelan	Powązkowska 74, 01-728 Warszawa	8.66
Carrefour Express	Księdza Juliana Chrościckiego 17, 01-999 Warszawa	7.96

Każdy ze sklepów spożywczych składa zamówienie do centrali sieci magazynów (przez e-mail, telefon, lub specjalną aplikację) raz w tygodniu. Każdy sklep może być obsługiwany przez dowolny magazyn, lub kilka magazynów. Ilość zamawianego towaru wynika z aktualnego stanu zapasów w magazynie przysklepowym i prognozy sprzedaży (wyniki modelu optymalizacyjnego są wartością orientacyjną, ale pozwalają podjąć lepszą decyzję, z których magazynów są sprowadzane produkty).

Raz w roku (jesienią) producenci dostarczają towar do magazynów. Ilość towaru jest wyliczana na podstawie oddzielnie przeprowadzonych obliczeń, zgodnych z prognozowanymi zapotrzebowaniem (patrz model optymalizacyjny)

## 2. Model BPMN



### 3. Model Optymalizacyjny

#### 3.1 Opis modelu

Model składa się z:

- Zbiorów:
  - Producentów
  - Magazynów
  - Produktów
  - Sklepów
- Parametrów:
  - Zapotrzebowanie
  - Pojemność magazynów
  - Kosztów za kilometr
  - Tygone w roku
  - Dystans od producentów do magazynów
  - Dystans od magazynów do sklepów
  - Tygodniowa sprzedaż
  - Pojemność magazynów w sklepach
- Zmiennych decyzyjnych:
  - Cotygodniowy zapas warzyw w magazynach sklepowych
  - Roczny transport od producentów do magazynów
  - Cotygodniowy transport od magazynów do sklepów
- Ograniczeń:
  - Roczny transport do magazynu nie może przekroczyć maksymalnej pojemności magazynu
  - Roczny transport do magazynu nie może przekroczyć zapotrzebowania
  - Suma dostaw do magazynu nie może być mniejsza niż transportów z tych magazynów do sklepów
  - Ustalenie minimalnego zapasu względem sprzedaży
  - Ustalenie zapotrzebowania na pierwszy tydzień
  - Ustalenie zapotrzebowania na kolejne tygodnie
  - Ilość towaru w sklepie nie może przekroczyć pojemności jego magazynu
- Funkcji celu:
  - Minimalizacja łącznych kosztów operacji

## 3.2 Model w AMPL

```
set PRODUCENTCI;
set MAGAZYNY;
set PRODUKTY;
set SKLEPY;

param zapotrzebowanie{PRODUCENTCI,PRODUKTY};
param pojemnosc_magazynu{MAGAZYNY};
param koszt_km;
param tydzien;
param dystans_do_magazynu{MAGAZYNY,PRODUCENTCI};
param dystans_do_sklepu{MAGAZYNY,SKLEPY};
param tygodniowa_sprzedarz{1..tydzien, SKLEPY, PRODUKTY};
param pojemnosc_magazynu_sklepu{SKLEPY};

var tygodniowy_zapas_sklepu {1..tydzien,SKLEPY,PRODUKTY} >= 0;
var roczny_transport_do_magazynu {PRODUCENTCI,MAGAZYNY,PRODUKTY} >= 0;
var tygodniowy_transport_do_sklepu {1..tydzien,MAGAZYNY,SKLEPY,PRODUKTY} >= 0;

minimize Laczny_koszt_operacji:
    sum {m in MAGAZYNY, s in SKLEPY, v in PRODUKTY, n in 1..tydzien}
        dystans_do_sklepu[m,s] * koszt_km * tygodniowy_transport_do_sklepu[n,m,s,v]
    +
    sum {p in PRODUCENTCI, m in MAGAZYNY, v in PRODUKTY}
        dystans_do_magazynu[m,p] * koszt_km * roczny_transport_do_magazynu[p,m,v];

subject to Maksymalna_pojemnosc_magazynu{m in MAGAZYNY}:
    sum {p in PRODUCENTCI, v in PRODUKTY} roczny_transport_do_magazynu[p,m,v] <= pojemnosc_magazynu[m];

subject to Warunek_zapotrzebowania{p in PRODUCENTCI, v in PRODUKTY}:
    sum {m in MAGAZYNY} roczny_transport_do_magazynu[p,m,v] <= zapotrzebowanie[p, v];

subject to Sprawdzenie_sumy_dostaw {m in MAGAZYNY, v in PRODUKTY}:
    sum {p in PRODUCENTCI} roczny_transport_do_magazynu[p, m, v] >= sum {s in SKLEPY, n in 1..tydzien} tygodniowy_transport_do_sklepu[n, m, s, v];

subject to Minimalny_zapas {s in SKLEPY, n in 1..tydzien, v in PRODUKTY}:
    tygodniowy_zapas_sklepu[n, s, v] >= 0.1 * tygodniowa_sprzedarz[n, s, v];

subject to Zapotrzebowanie_sklepu_pierszy_tydzien {s in SKLEPY, v in PRODUKTY}:
    tygodniowy_zapas_sklepu[1, s, v] = - tygodniowa_sprzedarz[1, s, v] + sum {m in MAGAZYNY} tygodniowy_transport_do_sklepu[1, m, s, v];

subject to Zapotrzebowanie_sklepu {s in SKLEPY, n in 2..tydzien, v in PRODUKTY}:
    tygodniowy_zapas_sklepu[n, s, v] = tygodniowy_zapas_sklepu[n-1, s, v] - tygodniowa_sprzedarz[n, s, v] + sum {m in MAGAZYNY} tygodniowy_transport_do_sklepu[n, m, s, v];

subject to Pojemnosc_magazynu_sklepowego {s in SKLEPY, n in 1..tydzien}:
    sum {v in PRODUKTY} tygodniowy_zapas_sklepu[n, s, v] + sum{v in PRODUKTY, m in MAGAZYNY} tygodniowy_transport_do_sklepu[n, m, s, v] <= pojemnosc_magazynu_sklepu[s];
```

## 4. Wyniki opracji

Po zaprojektowaniu, zaprogramowaniu i uruchomieniu modelu. Dla ustalonych danych wejściowych wyliczony minimalny koszt wynosi: 301641 zł.